



Title	SORD-M223を用いた9600BPS TSS端末プログラム
Author(s)	品川, 宏; 那須, 三郎
Citation	大阪大学大型計算機センターニュース. 1985, 59, p. 39-45
Version Type	VoR
URL	https://hdl.handle.net/11094/65669
rights	
Note	

The University of Osaka Institutional Knowledge Archive : OUKA

<https://ir.library.osaka-u.ac.jp/>

The University of Osaka

SORD-M223を用いた9600BPS TSS端末プログラム

大阪大学基礎工学部

品 川 宏 ・ 那 須 三 郎

1. はじめに

パーソナルコンピュータをTSS端末として使用する際、その通信速度は電話を利用する場合では音響カップラーの性能によって、300又は、1200BPSと決まります。一方、専用回線を利用する場合では、9600BPSまで通信速度をアップさせることが可能です。我々は、SORD-M223を9600BPS用端末として使用できるプログラムを作製したのでここに紹介します。

プログラムはSORDのCommunication BASIC言語(TBASIC)中のアセンブラ言語によって作製されたプログラムを用い、XON・XOFFコントロール機能を持つために9600BPSの通信速度にも対応できます。また、以下の機能を持つためTSS端末としても便利なものです。

1) SORD, ACOS間でファイルの転送ができる。

2) プリンター・プロッターを直結できる。

実際、TSSを行う場合、特にファイルの転送に際して、9600BPSの通信速度は非常に効率よい会話を生み出します。

2. システム構成

システム構成を図1に示します。使用したモデムはMicro 400です。音響カップラーによる通信(300, 1200BPS)も何らプログラムを変更することなく行えます。SORD-M223にはRS-232Cのポートが

2ヶとフロッピー・ディスク・ドライブ(FDD)が2ヶ、本体に内蔵されているので付け加える必要のある周辺機器は、モデムとプリンターあるいはプロッター程度と考えています。




図1 システム構成

3. プログラムについて

本プログラムはSORD社TBASICをオペレーティング・システムとして用い、アセンブラ言語で作製されたプログラムをCALL #でサブルーチン・プログラムとして利用しています。その

詳細はTBASICのマニュアルに譲るとして、ここでは使用しているサブルーチン・プログラムの機能のみ示しておきます。

CALL#13,0 ……初期設定
 CALL#13,1 ……クローズ
 CALL#13,2 ……キー入力
 CALL#13,3 ……キャラクタ表示
 CALL#13,4 ……シリアル入力(HOSTコンピュータからの入力)
 CALL#13,5 ……シリアル出力(HOSTコンピュータへの出力)
 CALL#13,6 ……ブレイク・キャラクタ送信
 CALL#13,7 ……ブレイク・キャラクタ送信停止
 CALL#13,13 ……受信バッファ指定

このプログラムの特色であるXON・XOFFコントロール機能について説明します。

受信バッファとして使用するために、DIM A1(n)で次元配列エリアを確保します。このときバッファ・サイズは $(n+1) \times 4$ バイトになります。受信バッファの空きエリアが32バイトになるとXOFFを送信して、送信の一時停止を要求し、受信バッファ内のデータが32バイトになるとXONを送信して、送信の再開を要求します(図2)。バッファ・サイズを決めるnはできるだけ大きくとりま

す。また、このコントロール機能以外にも要

所で、XON(S0\$), XOFF(S1\$)を送信しています。




図2 XON・XOFF送信時の受信バッファの状態

4. モード選択とその使用法

ESC キーを押すことにより、以下の0~7でモード選択できます。

- 0) ESC + 0 ……モード選択の状態から脱出する。
- 1) ESC + 1 ……ブレイク・キャラクタを送信する。

- 2) ESC + 2 …… SORDのファイルをACOSへ送信する。
- 3) ESC + 3 …… ACOSからデータ・ファイルを受信する。
- 4) ESC + 4 …… ACOSからプログラム・ファイルを受信する。
- 5) ESC + 5 …… プリンターあるいはプロッターを使用状態にする。
- 6) ESC + 6 …… TSSプログラムを終る。
- 7) ESC + 7 …… 画面消去を行う。

3), 4), 5) は再度 ESC キーを押すことにより、その他は終了とともに、それぞれのモードから脱出できます。次に、具体的な使用法を述べます。

0) エスケープ・コードからの脱出

- ・間違って ESC キーを押したときに使います。

1) ブレイク

- ・ 300,1200BPSでの会話であれば ESC + 1 と押します。
- ・ 9600BPSでの会話であれば HALT キーを押して下さい。

2) ファイルの送信

- ・ SORDより転送したいファイルの入ったディスクットをFDD#1にセットします。
- ・ エディターを起動させ、I コマンドの後 ESC + 2 と押します。
- ・ ファイル名を入力し、そのファイルをオープンします。
- ・ 転送終了後、ファイルはクローズされ、メッセージが表示されます。

3) データ・ファイルの受信

- ・ SORDのFDD#1にディスクットをセットします。
- ・ ESC + 3 と押した後、SORDにつくるファイル名を入力します。
- ・ \$FLIST_Uファイル名(ACOS)_UW = 100, 1 で転送開始です。
- ・ 転送終了後、ESC キーを押して下さい。

4) プログラム・ファイルの受信

- ・ 3) と同様の操作をして下さい。

5) プリンター・プロッター

- ・ ESC + 5 と押した後、80字と132字の選択をして下さい。
- ・ キー入力及びACOSから受信した内容の両方をプリンターへ出力します。
- ・ 再度、ESC キーを押すともとの状態に戻ります。

6) プログラムの終了

- ・ `ESC` + 6 と押せば、SORD は TSS 端末から開放されます。

7) 画面消去

- ・ `ESC` + 7 と押せば、画面消去を行ないます。

8) バック・スペース

- ・ `DEL` キーは効かないので `←` キーを押して下さい。

5. おわりに

実際に通信を処理しているアセンブラ言語によるプログラムの内容は公表されていないので、ここに載せるプログラム・リストは BASIC の部分のみです。但し、XON・XOFF コントロール機能は SORD 社 TBASIC のうち Rev. 01 J 以降のバージョンにのみ備わっています。また、このプログラムではファイルの送信を ACOS のエディターのビルト・モードで※が来ると一行送るという方法を採用していますが、READ PPT を使用してファイル転送を行いたい時には簡単な修正で実行できます。

なお、このプログラムの必要な方は御連絡下さい。コピーサービス致します。

```

100 ON ERROR GOTO 9000
200 DIM A$(1), A$(1), F$(13)
210 DIM A1(4096)
220 LET S0$ = CHR$(17)
: LET S1$ = CHR$(19)
290 PRINT CLEAR
300 PRINT CURSOR(18,20); " *****      *****      +++++ "
310 PRINT CURSOR(18,19); " *****      *****      "
320 PRINT CURSOR(18,18); " ***      ***      "
330 PRINT CURSOR(18,17); " ***      ***      "
340 PRINT CURSOR(18,16); " ***      ***      "
350 PRINT CURSOR(18,15); " ***      ***      "
360 PRINT CURSOR(18,14); " ***      ***      "
370 PRINT CURSOR(18,13); " ***      ***      "
380 PRINT CURSOR(18,12); " ***      ***      "
390 PRINT CURSOR(18,11); " ***      ***      "
400 PRINT
410 PRINT CURSOR(20,8); " Originally programed by Saburo Hasu "
420 PRINT CURSOR(20,7); " Modified for 9600 BAUD by H.Shinagawa "
450 PRINT CHR$(13); "SET HANDSET & STRIKE ANY KEY";
: INPUT #0, A$
500 PRINT CHR$(13); "START OF TSS PROGRAM"
: LET T0 = TIME
600 CALL #13, 0, 0, 2, 0, 0, 0, 13
650 CALL #13, 13, A1, 1
700 PRINT CHR$(13); "TERMINAL MODE"; CHR$(13)
750 CALL #13, 5, S0$
800 CALL #13, 2, A$, 900
: IF A$ = CHR$(27) THEN GOTO 1000
850 CALL #13, 5, A$
900 CALL #13, 4, A$, 800
950 CALL #13, 3, A$
980 GOTO 800
1000 REM*****ESC MODE*****
1050 CALL #13, 5, S1$
1100 PRINT
: PRINT
1200 PRINT " MODE SELECTION"
1250 PRINT
1300 PRINT ": 0 : OUTPUT OF ESCAPE CODE : 1 : OUTPUT OF BREAK CHARACTER"
1400 PRINT ": 2 : TRANSMISSION OF FILE : 3 : RECEIVING OF DATA FILE"
1500 PRINT ": 4 : RECEIVING OF PROG FILE : 5 : LIST USING PRINTER"
1600 PRINT ": 6 : END OF OPERATION : 7 : DISPLAY CLEAR"
1700 PRINT
1800 INPUT "YOUR SELECTION ? "; S
: IF S = 0 THEN GOTO 750
: IF S > 7 THEN GOTO 1700
1900 ON S GOTO 3000, 4000, 5000, 6000, 7000, 8000, 2000
2000 REM*****CLEAR*****
2100 PRINT CLEAR
2200 PRINT
2300 GOTO 700
3000 REM*****BREAK*****
3100 PRINT CHR$(13); "OUTPUT OF BREAK CHARACTER"
3200 CALL #13, 6
: SLEEP 1
: CALL #13, 7
3400 PRINT
: PRINT "When 9600 BAUD --> Hit the (HALT) KEY and (CR) !! "
3900 PRINT
: GOTO 700
4000 REM*****FILE TRANSMISSION*****
4050 CALL #13, 5, S0$
4100 PRINT CHR$(13); "TRANSMISSION OF FILE"
: INPUT "FILE NAME OF YOUR DISK ? ", F$
: OPEN "1:" + F$ AS FILE 1
4200 PRINT "*";
4300 READ #1, A$
4400 CALL #13, 3, A$
4500 CALL #13, 5, A$
4550 IF A$ = CHR$(13) THEN
: GOSUB 10000
4600 GOTO 4300
4900 CALL #13, 5, S1$
: CLOSE 1
: PRINT
: GOTO 700

```

```

5000 REM*****FILE RECEIVING*****
5100 CALL #13, 5, 50$
5200 PRINT CHR$(13); "RECEIVING OF DATA FILE (ACOS FILE TO YOUR DISK) "
5300 INPUT "NAME OF FILE STORED IN YOUR DISK : ? ", F$
: OPEN "1:"*F$ AS FILE 1
5400 CALL #13, 4, A$, 5600, 127, 5400
5500 CALL #13, 3, A$
: WRITE #1, A$;
: GOTO 5400
5600 CALL #13, 2, A$, 5400
: IF A$ = CHR$(27) THEN GOTO 5900
5700 CALL #13, 5, A$
: GOTO 5600
5900 CALL #13, 5, S1$
: CLOSE 1
: PRINT
: PRINT
: PRINT "***** (SORD) HAS RECEIVED <" ; F$ ; "> !! *****"
5950 PRINT
: GOTO 700
6000 REM*****FILE RECIVING (PROG FILE)*****
6100 CALL #13, 5, 50$
6200 PRINT CHR$(13); "RECEIVING OF PROG FILE (ACOS FILE TO YOUR DISK) "
6300 INPUT "NAME OF FILE STORED IN YOUR DISK : ? ", F$
: OPEN "1:"*F$ AS FILE 1
6400 CALL #13, 4, A$, 6600, 127, 6400
6500 CALL #13, 3, A$
: WRITE #1, A$;
: GOTO 6400
6600 CALL #13, 2, A$, 6400
: IF A$ = CHR$(27) THEN GOTO 6900
6700 CALL #13, 5, A$
: GOTO 6600
6900 CALL #13, 5, S1$
: CLOSE 1
: PRINT
: PRINT
: PRINT "***** (SORD) HAS RECEIVED <" ; F$ ; "> !! *****"
6950 PRINT
: GOTO 700
7000 REM*****PRINTER*****
7050 CALL #13, 5, 50$
7100 OPEN "SOUTA" FOR OUTPUT AS FILE 1 MODE 3
7150 PRINT
: PRINT CHR$(13), "LENGTH OF THE LIST IN PRINTER"
7200 INPUT "80 ( N ) OR 132 ( Y ) ? : " ; A9$
7210 PRINT #1, CHR$(10)
7230 IF A9$ = "Y" THEN GOTO 7300
7250 IF A9$ = "N" THEN GOTO 7350
7280 GOTO 7200
7300 PRINT #1, CHR$(20)
: PRINT #1, CHR$(15)
: GOTO 7400
7350 PRINT #1, CHR$(18)
: PRINT #1, CHR$(14)
7400 CALL #13, 4, A$, 7700
7500 CALL #13, 3, A$
: IF A$ = CHR$(10) THEN
: LET A$ = CHR$(32)
7650 PRINT #1 ; A$ ;
7700 CALL #13, 2, A$, 7400
7710 IF A$ = CHR$(27) THEN GOTO 7800
7720 PRINT #1 ; A$ ;
7750 CALL #13, 5, A$
7780 GOTO 7700
7800 PRINT #1, CHR$(10)
: IF A9$ = "Y" THEN
: PRINT #1, CHR$(20)
: ELSE
: PRINT #1, CHR$(18)
7900 CALL #13, 5, S1$
: CLOSE 1
: PRINT
: GOTO 700
8000 REM*****END*****
8010 REM*** TOKI ***
8020 LET T1 = TIME
: LET T2 = INT((T1-T0)/60)
: LET T3 = T1-T0-T2*60
8100 CALL #13, 1
: PRINT
: PRINT "END OF OPERATION"
: PRINT
: PRINT CHR$(13), "USED TIME : " ; T2 ; " MIN " ; T3 ; " SEC"
8200 END

```

```

9000 REM*** ERROR ***
9050 LET I0 = ERR
: IF I0 = 11 THEN
: PRINT "***END OF YOUR DISK FILE " ; F$ ; " *****"
: GOTO 4900
9100 IF I0 = 4 THEN
: PRINT "YOUR DISKET WAS FILLED. USE NEW DISKET."
: GOTO 9400
9200 IF I0 = 84 THEN
: PRINT "FILE " ; F$ ; " IS ALREADY EXISTED. USE NEW NAME."
: GOTO 9400
9250 LET E0 = ERRL
: IF E0 = 1800 THEN GOTO 1700
9280 IF E0 = 1900 THEN GOTO 1700
9300 GOTO 9500
9400 RESUME 700
9500 PRINT CHR$(113) ; "ERROR : " ; ERR , "ERROR STATEMENT NO. : " ; ERRL
9600 CALL #13 , 1
9700 END
10000 REM*****WAIT FOR "*" FROM ACOS EDITOR*****
10010 PRINT CHR$(113) ;
10100 CALL #13 , 4 , A$ , 10200 , 42 , 10300
10200 CALL #13 , 2 , A$
10210 IF A$ = CHR$(27) THEN GOTO 10400
10220 GOTO 10100
10300 CALL #13 , 3 , A$
10400 RETURN
10500 END

```